
ElixirRetry
Simple Elixir macros for linear retry, exponential backoff and wait with composable delays.

  
    
  
  Installation


Add retry to your list of dependencies in mix.exs:
  def deps do
    [{:retry, "~> 0.18"}]
  end
Ensure retry is started before your application:
  def application do
    [applications: [:retry]]
  end

  
    
  
  Documentation


Check out the API reference for the latest documentation.

  
    
  
  Features



  
    
  
  Retrying


The retry([with: _,] do: _, after: _, else: _) macro provides a way to retry a block of code on failure with a variety of delay and give up behaviors. By default, the execution of a block is considered a failure if it returns :error, {:error, _} or raises a runtime error.
Both the values and exceptions that will be retried can be customized. To control which values will be retried, provide the atoms option. To control which exceptions are retried, provide the rescue_only option. For example:
retry with: ..., atoms: [:not_ok], rescue_only: [CustomError] do
  ...
end
Both atoms and rescue_only can accept a number of different types:
	An atom (for example: :not_okay, SomeStruct, or CustomError). In this case, the do block will be retried in any of the following cases:	The atom itself is returned
	The atom is returned in the first position of a two-tuple (for example, {:not_okay, _})
	A struct of that type is returned/raised


	The special atom :all. In this case, all values/exceptions will be retried.
	A function (for example: fn val -> String.starts_with?(val, "ok") end) or partial function (for example: fn {:error, %SomeStruct{reason: "busy"}} -> true). The function will be called with the return value and the do block will be retried if the function returns a truthy value. If the function returns a falsy value or if no function clause matches, the do block will not be retried.
	A list of any of the above. The do block will be retried if any of the items in the list matches.

The after block evaluates only when the do block returns a valid value before timeout. On the other hand, the else block evaluates only when the do block remains erroneous after timeout. Both are optional. By default, the else clause will return the last erroneous value or re-raise the last exception. The default after clause will simply return the last successful value.
Example -- constant backoff
result = retry with: constant_backoff(100) |> Stream.take(10) do
  ExternalApi.do_something # fails if other system is down
after
  result -> result
else
  error -> error
end
This example retries every 100 milliseconds and gives up after 10 attempts.
Example -- linear backoff
result = retry with: linear_backoff(10, 2) |> cap(1_000) |> Stream.take(10) do
  ExternalApi.do_something # fails if other system is down
after
  result -> result
else
  error -> error
end
This example increases the delay linearly with each retry, starting with 10 milliseconds, caps the delay at 1 second and gives up after 10 attempts.
Example -- exponential backoff
result = retry with: exponential_backoff() |> randomize |> expiry(10_000), rescue_only: [TimeoutError] do
  ExternalApi.do_something # fails if other system is down
after
  result -> result
else
  error -> error
end
Example -- optional clauses
result = retry with: constant_backoff(100) |> Stream.take(10) do
  ExternalApi.do_something # fails if other system is down
end
This example is equivalent to:
result = retry with: constant_backoff(100) |> Stream.take(10) do
  ExternalApi.do_something # fails if other system is down
after
  result -> result
else
  e when is_exception(e) -> raise e
  e -> e
end
Example -- retry annotation
use Retry.Annotation

@retry with: constant_backoff(100) |> Stream.take(10)
def some_func(arg) do
  ExternalApi.do_something # fails if other system is down
end
This example shows how you can annotate a function to retry every 100 milliseconds and gives up after 10 attempts.
Delay streams
The with: option of retry accepts any Stream that yields integers. These integers will be interpreted as the amount of time to delay before retrying a failed operation. When the stream is exhausted retry will give up, returning the last value of the block.
Example
result = retry with: Stream.cycle([500]) do
  ExternalApi.do_something # fails if other system is down
after
  result -> result
else
  error -> error  
end
This will retry failures forever, waiting 0.5 seconds between attempts.
Retry.DelayStreams provides a set of fully composable helper functions for building useful delay behaviors such as the ones in previous examples. See the Retry.DelayStreams module docs for full details and addition behavior not covered here. For convenience these functions are imported by use Retry so you can, usually, use them without prefixing them with the module name.

  
    
  
  Waiting


Similar to retry(with: _, do: _), the wait(delay_stream, do: _, after: _, else: _) macro provides a way to wait for a block of code to be truthy with a variety of delay and give up behaviors. The execution of a block is considered a failure if it returns false or nil.
wait constant_backoff(100) |> expiry(1_000) do
  we_there_yet?
after
  _ ->
    {:ok, "We have arrived!"}
else
  _ ->
    {:error, "We're still on our way :("}
end
This example retries every 100 milliseconds and expires after 1 second.
The after block evaluates only when the do block returns a truthy value. On the other hand, the else block evaluates only when the do block remains falsy after timeout. Both are optional. By default, a success value will be returned as {:ok, value} and an erroneous value will be returned as {:error, value}.
Pretty nifty for those pesky asynchronous tests and building more reliable systems in general!

Provides a convenient interface to retrying behavior. All durations are
specified in milliseconds.
Examples
use Retry
import Stream

retry with: exponential_backoff |> randomize |> cap(1_000) |> expiry(10_000) do
# interact with external service
end

retry with: linear_backoff(10, 2) |> cap(1_000) |> take(10) do
# interact with external service
end

retry with: cycle([500]) |> take(10) do
# interact with external service
end
The first retry will exponentially increase the delay, fudging each delay up
to 10%, until the delay reaches 1 second and then give up after 10 seconds.
The second retry will linearly increase the retry by a factor of 2 from 10ms giving up after 10 attempts.
The third example shows how we can produce a delay stream using standard
Stream functionality. Any stream of integers may be used as the value of
with:.

    Functions
  


    
      
        retry(opts, clauses)

      


        Retry a block of code delaying between each attempt the duration specified by
the next item in the with delay stream.



    


    
      
        retry_while(args, arg2)

      


        Retry a block of code until halt is emitted delaying between each attempt
the duration specified by the next item in the with delay stream.



    


    
      
        wait(stream_builder, clauses)

      


        Wait for a block of code to be truthy delaying between each attempt
the duration specified by the next item in the delay stream.

      retry(opts, clauses)

Retry a block of code delaying between each attempt the duration specified by
the next item in the with delay stream.
Both the values and exceptions that will be retried can be customized. To control which values
will be retried, provide the atoms option. To control which exceptions are retried, provide
the rescue_only option. For example:
retry with: ..., atoms: [:not_ok], rescue_only: [CustomError] do
  ...
end
Both atoms and rescue_only can accept a number of different types:
	An atom (for example: :not_okay, SomeStruct, or CustomError). In this case, the do
block will be retried in any of the following cases:	The atom itself is returned
	The atom is returned in the first position of a two-tuple (for example, {:not_okay, _})
	A struct of that type is returned/raised


	The special atom :all. In this case, all values/exceptions will be retried.
	A function (for example: fn val -> String.starts_with?(val, "ok") end) or partial function
(for example: fn {:error, %SomeStruct{reason: "busy"}} -> true). The function will be called
with the return value and the do block will be retried if the function returns a truthy value.
If the function returns a falsy value or if no function clause matches, the do block
will not be retried.
	A list of any of the above. The do block will be retried if any of the items in the list
matches.

The after block evaluates only when the do block returns a valid value before timeout.
On the other hand, the else block evaluates only when the do block remains erroneous after timeout.
Example
use Retry

retry with: exponential_backoff() |> cap(1_000) |> expiry(1_000), rescue_only: [CustomError] do
  # interact with external service
after
  result -> result
else
  error -> error
end
The after and else clauses are optional. By default, a successful value is just returned. If
the timeout expires, the last erroneous value is returned or the last exception is re-raised.
Essentially, this:
retry with: ... do
  ...
end
Is equivalent to:
retry with: ... do
  ...
after
  result -> result
else
  e when is_exception(e) -> raise e
  e -> e
end

Retry a block of code until halt is emitted delaying between each attempt
the duration specified by the next item in the with delay stream.
The return value for block is expected to be {:cont, result}, return
{:halt, result} to end the retry early.
An accumulator can also be specified which might be handy if subsequent
retries are dependent on the previous ones.
The initial value of the accumulator is given as a keyword argument acc:.
When the :acc key is given, its value is used as the initial accumulator
and the do block must be changed to use -> clauses, where the left side
of -> receives the accumulated value of the previous iteration and
the expression on the right side must return the :cont/:halt tuple
with new accumulator value as the second element.
Once :halt is returned from the block, or there are no more elements,
the accumulated value is returned.
Example
retry_while with: linear_backoff(500, 1) |> take(5) do
  call_service
  |> case do
    result = %{"errors" => true} -> {:cont, result}
    result -> {:halt, result}
  end
end
Example with acc:
retry_while acc: 0, with: linear_backoff(500, 1) |> take(5) do
  acc ->
    call_service
    |> case do
      %{"errors" => true} -> {:cont, acc + 1}
      result -> {:halt, result}
    end
end

  



  
    
      
    
    
      wait(stream_builder, clauses)

Wait for a block of code to be truthy delaying between each attempt
the duration specified by the next item in the delay stream.
The after block evaluates only when the do block returns a truthy
value. On the other hand, the else block evaluates only when the
do block remains falsy after timeout.Both are optional. By default,
a success value will be returned as {:ok, value} and an erroneous
value will be returned as {:error, value}.
Example
wait linear_backoff(500, 1) |> take(5) do
  we_there_yet?
after
  _ ->
    {:ok, "We have arrived!"}
else
  _ ->
    {:error, "We're still on our way :("}
end

A @retry annotation that will retry the function according to the retry settings.
Examples
use Retry.Annotation

@retry with: constant_backoff(100) |> Stream.take(10)
def some_func(arg) do
  # ...
end

        on_definition(env, kind, name, args, guards, body)

      on_definition(env, kind, name, args, guards, body)



        
          
        

    

  


  


  


        

      


  

    
Retry.DelayStreams 
    



      
This module provide a set of helper functions that produce delay streams for
use with retry.

        factor()

      


    





  
    Functions
  


    
      
        cap(delays, max)

      


        Returns a stream that is the same as delays except that the delays never
exceed max. This allow capping the delay between attempts to some max value.



    


    
      
        constant_backoff(delay \\ 100)

      


        Returns a constant stream of delays.



    


    
      
        expiry(delays, time_budget, min_delay \\ 100)

      


        Returns a delay stream that is the same as delays except it limits the total
life span of the stream to time_budget. This calculation takes the execution
time of the block being retried into account.



    


    
      
        exponential_backoff(initial_delay \\ 10, factor \\ 2)

      


        Returns a stream of delays that increase exponentially.



    


    
      
        jitter(delays)

      


        Returns a stream in which each element of delays is randomly adjusted to a number
between 1 and the original delay.



    


    
      
        linear_backoff(initial_delay, factor)

      


        Returns a stream of delays that increase linearly.



    


    
      
        randomize(delays, proportion \\ 0.1)

      


        Returns a stream in which each element of delays is randomly adjusted no
more than proportion of the delay.

          @type factor() :: pos_integer() | float()

          @spec cap(Enumerable.t(), pos_integer()) :: Enumerable.t()


      


Returns a stream that is the same as delays except that the delays never
exceed max. This allow capping the delay between attempts to some max value.
Example
retry with: exponential_backoff() |> cap(10_000) do
  # ...
end
Produces an exponentially increasing delay stream until the delay reaches 10
seconds at which point it stops increasing

  



    

  
    
      
    
    
      constant_backoff(delay \\ 100)



        
          
        

    

  


  

      

          @spec constant_backoff(pos_integer()) :: Enumerable.t()


      


Returns a constant stream of delays.
Example
retry with: constant_backoff(50) do
  # ...
end

  



    

  
    
      
    
    
      expiry(delays, time_budget, min_delay \\ 100)



        
          
        

    

  


  

      

          @spec expiry(Enumerable.t(), pos_integer(), pos_integer()) :: Enumerable.t()


      


Returns a delay stream that is the same as delays except it limits the total
life span of the stream to time_budget. This calculation takes the execution
time of the block being retried into account.
The execution of the code within the block will not be interrupted, so
the total time of execution may run over the time_budget depending on how
long a single try will take.
Optionally, you can specify a minimum delay so the smallest value doesn't go
below the threshold.
Example
retry with: exponential_backoff() |> expiry(1_000) do
  # ...
end
Produces a delay stream that ends after 1 second has elapsed since its
creation.

  



    

    

  
    
      
    
    
      exponential_backoff(initial_delay \\ 10, factor \\ 2)



        
          
        

    

  


  

      

          @spec exponential_backoff(pos_integer(), factor()) :: Enumerable.t()


      


Returns a stream of delays that increase exponentially.
Example
retry with: exponential_backoff do
  # ...
end

  



  
    
      
    
    
      jitter(delays)



        
          
        

    

  


  

      

          @spec jitter(Enumerable.t()) :: Enumerable.t()


      


Returns a stream in which each element of delays is randomly adjusted to a number
between 1 and the original delay.
Example
retry with: exponential_backoff() |> jitter() do
  # ...
end

  



  
    
      
    
    
      linear_backoff(initial_delay, factor)



        
          
        

    

  


  

      

          @spec linear_backoff(pos_integer(), factor()) :: Enumerable.t()


      


Returns a stream of delays that increase linearly.
Example
retry with: linear_backoff(50, 2) do
  # ...
end

  



    

  
    
      
    
    
      randomize(delays, proportion \\ 0.1)



        
          
        

    

  


  

      

          @spec randomize(Enumerable.t(), float()) :: Enumerable.t()


      


Returns a stream in which each element of delays is randomly adjusted no
more than proportion of the delay.
Example
retry with: exponential_backoff() |> randomize do
  # ...
end
Produces an exponentially increasing delay stream where each delay is randomly
adjusted to be within 10 percent of the original value

